

Green Application Placement in the Cloud-IoT Continuum

Stefano Forti^[0000–0002–4159–8761] and Antonio Brogi^[0000–0003–2048–2468]

Department of Computer Science, University of Pisa, Italy
{stefano.forti, antonio.brogi}@di.unipi.it

Abstract. Green software engineering aims at reducing the environmental impact due to developing, deploying, and managing software systems. Meanwhile, Cloud-IoT paradigms can contribute to improving energy and carbon efficiency of application deployments by (i) reducing the amount of data and the distance they must travel across the network, (ii) by exploiting idle edge devices to support application deployment. In this article, we propose a declarative methodology and its Prolog prototype for determining placements of application services onto Cloud-IoT infrastructures so to optimise energy and carbon efficiency, also considering different infrastructure power sources and operational costs. The proposal is assessed over a motivating example.

1 Introduction

ICT energy demand could possibly reach 14% of the total worldwide footprint by 2040 [3]. As climate scientists agree on the urgency of reducing the human impact on the environment, *green software engineering* is getting increasing attention as a possible way to contain ICT energy usage and carbon emissions, through achieving a more sustainable software life-cycle [8]. While much work has focussed on embedding sustainability principles in software design phases, less work has proposed methodologies and tools to improve and assess software lifecycle sustainability [20], i.e. from application testing to deployment and runtime management. Meanwhile, Cloud-IoT computing paradigms – e.g. Fog, Edge computing [7] – have been proposed to improve on the Quality of Service (QoS) of emerging latency-sensitive and bandwidth-hungry applications. As highlighted by some authors [22,17], those paradigms can also represent greener alternatives to the Cloud paradigm as they can exploit pervasive and possibly idle computational devices closer to the IoT, thus improving on energy efficiency of those idle resources and reducing unnecessary data transfer from/to the Cloud.

To achieve the above sustainability goals, it is crucial to place application services so to meet all their requirements *and* by determining the best trade-off between the operation costs of their deployment and the expected energy consumption and carbon emissions, which very much depend on the characteristics of the target deployment nodes (i.e. energy profile, power sources, power usage effectiveness). While the problem of placing application services onto Cloud-IoT infrastructure to meet their software, hardware, IoT and network requirements has been extensively studied [18,5], the problem of determining energy- and carbon-aware placements was only marginally addressed until very recently [1].

In this article, based on our previous work [10], we illustrate a declarative programming solution to the problem of determining energy- and carbon-aware application placements in Cloud-IoT settings, also capable of estimating operational deployment costs. Our methodology permits determining application placements that meet software, hardware, IoT, latency and bandwidth requirements, and to estimate deployment costs, energy consumption and carbon emissions, relying on data disclosed on the available Cloud-IoT nodes. Being declarative, our approach is easy to understand and to extend, e.g. by employing alternative formulas to estimate all of the above. A Prolog open-source prototype, **GFogBrain**, is assessed over a motivating example based on lifelike data.

The rest of this article is organised as follows. Section 2 describes the model and methodology of **GFogBrain**, while showcasing its functioning over a lifelike motivating example. Section 3 briefly discusses some closely related work, and Section 4 concludes by pointing to some directions for future work.

2 **GFogBrain** in Action

In this section, we illustrate **GFogBrain**'s prototype and methodology by means of a lifelike motivating example from *smart environments* [4]. **GFogBrain** extends our previous work in the field of context- and QoS-aware placements of Cloud-IoT applications, to also determine energy- and carbon-aware placements. Particularly, we extend the model prototype of [10] to consider all necessary ingredients to estimate energy consumption and carbon emissions of running applications, and operational costs (i.e. due to leasing computational resources to keep application services up and running).

Our goal is to support application operators, enabling them to informedly identify placements that can reduce energy consumption and carbon emissions, while assessing the impact that *being greener* could have on the operational costs of their deployments. In the next paragraphs, we detail our declarative application and infrastructure model and the declarative programming methodology implemented by **GFogBrain** to achieve such a goal.

Applications Requirements – As in [10], application A made of services $S_1 \dots S_k$ is declared as

```
application(A, [S1, ..., Sk]).
```

The software, hardware¹ and IoT requirements of service S are declared as

```
service(S, SoftwareReqs, HardwareReqs, IoTReqs).
```

Finally, interactions between services S_1 and S_2 with associated maximum end-to-end latency and minimum bandwidth requirements are declared as

```
s2s(S1, S2, MaxLatency, MinBandwidth).
```

¹ For the sake of simplicity, we represent hardware units as integers as in [10].

Example 1. The application of Fig. 1 consists of two interacting services – Lights Driver and ML Optimiser – for optimising ambient lighting in a museum based on processing real-time video footage. The Lights Driver requires 2GB and Ubuntu to run, and to reach out a video-camera and a lights hub. Similarly, ML Optimiser requires 16GB of RAM and the availability of Ubuntu, MySQL and Python on the deployment node, which must also be equipped with a GPU for processing streamed data. Finally, the interaction from the Lights Driver to the ML Optimiser require at least 16Mbps of available bandwidth and tolerates at most 20 ms latency. On the other hand, the the interaction from the ML Optimiser to the Lights Driver needs only 0.5 Mbps, with a latency lower than 50 ms. Such an application can be simply declared as in Fig. 2. □




Fig. 1. Example application.

```
application(lightsApp, [mlOptimiser, lightsDriver]).
  service(mlOptimiser, [mySQL, python, ubuntu], 16, [gpu]).
  service(lightsDriver, [ubuntu], 2, [videocamera, lightshub]).
  s2s(mlOptimiser, lightsDriver, 50, 0.5).
  s2s(lightsDriver, mlOptimiser, 20, 16).
```

Fig. 2. Example application declaration.

Infrastructure capabilities – Complementarily to application service requirements, Cloud-IoT nodes can be declared along with their software, free hardware and IoT capabilities, and with the unit hourly cost for leasing hardware resources:

```
node(NodeId, SoftwareCapabilities, FreeHW, IoTCapabilities).
cost(NodeId, UnitHWCostPerHour).
```

Similarly, end-to-end links between nodes N1 and N2 are declared, along with their FeaturedLatency and FeaturedBandwidth, as in

```
link(N1, N2, FeaturedLatency, FeaturedBandwidth).
```

The *power usage effectiveness* (PUE) associated to a node is the ratio between the overall energy needed for keeping the node working and the energy that the node uses for actual computation. For instance, a PUE of 1.5 indicates that for every 1kWh spent in computation, another 0.5kWh is needed for non-IT tasks (e.g. cooling, lighting, network) that keep the server working. Typical values of

the PUE range between 1.2 and 1.9. Extending the model of [10], we assume that node operators can disclose information about the total hardware (free and in use) at each node and the associated PUE as in

```
totHW(N,TotalHardware).                pue(N,PUE).
```

Node operators can then specify the energy consumption profile of each node via predicates like

```
energyProfile(N,Load,EnergyConsumption) :- ...
```

where `EnergyConsumption` is obtained in kWh as a, possibly non-linear, function of the current percentage `Load` at node `N`. Existing processors show a baseline energy consumption even when they are idle, which increases as the node workload increases [19].

Last, the percentages of the energy mix of each node can be specified as in

```
energySourceMix(N,[(P1,Source1), ..., (PK,SourceK)]).
```

where `PJ` is the percentage of electricity that node `N` receives from `SourceJ`.

We finally assume that average CO₂ emissions for each source are declared in a public knowledge base of facts like `emissions(Source, Mu)`, where `Mu` are the emissions in kgCO₂/kWh for `Source`, e.g. as those reported in Table 1.

Note that, when energy-related information is not disclosed, **GFogBrain** easily allows to employ default data or data taken from public audits such as [13].

Table 1. CO₂ emissions per power source [16].

Power Source	Emissions [kgCO ₂ /kWh]
<i>gas</i>	0.610
<i>coal</i>	1.100
<i>on shore wind</i>	0.0097
<i>off shore wind</i>	0.0165
<i>solar</i>	0.05

Example 2. Consider the Cloud-IoT infrastructure of Fig. 3 to deploy the application of Example 1. Fig. 4 epitomises the declaration of the capabilities and energy information of all three nodes. We only show the declaration of the link between **Private Cloud** and **Access Point**.

Note that, for instance, node **Private Cloud** currently features 128 free hardware units (out of the 150 totally available), each offered for 0.0016 cents per hour, and that its energy consumption in kWh is given by a function $\phi(L)$ of the current workload L such that $\phi(w) = 0.1\text{kWh} + 0.01 \cdot \log(L)$ with $L \in [0, 100]$. Besides, **Private Cloud** is powered by an energy mix coming for 30% from a solar plant and for 70% from a coal plant, as declared by `energySourceMix/2`. Last, the PUE of **Private Cloud** is 1.9. \square




Fig. 3. Example Cloud-IoT infrastructure.

```

node(privateCloud,[ubuntu, mySQL, python], 128, [gpu]).
  cost(privateCloud,0.0016).      toHW(privateCloud,150).
  energyProfile(privateCloud,L,E) :- E is 0.1 + 0.01*log(L).
  pue(privateCloud,1.9).
  energySourceMix(privateCloud,[(0.3,solar), (0.7,coal)]).
node(accesspoint,[ubuntu, mySQL, python], 4, [lightshub, videocamera]).
  cost(accesspoint,0.003).      toHW(accesspoint,6).
  energyProfile(accesspoint,L,E) :- E is 0.05 + 0.03*log(L).
  pue(accesspoint,1.5).
  energySourceMix(accesspoint,[(0.1,gas),(0.8,coal),(0.1,onshorewind)]).
node(edgenode,[ubuntu, python], 8, [gpu, lightshub, videocamera]).
  cost(edgenode,0.005).      toHW(edgenode,12).
  energyProfile(edgenode,L,E) :- L <= 50 -> E is 0.08; E is 0.1.
  pue(edgenode,1.2).
  energySourceMix(edgenode,[(0.5,coal), (0.5,solar)]).

link(privateCloud, accesspoint, 5, 1000).

```

Fig. 4. Example infrastructure declaration.

Energy-, Carbon- and Cost-aware Placements – Fig. 6 shows² how the **GFogBrain** prototype determines energy- and carbon-aware application placements in Cloud-IoT settings. Predicate `placements/2` (lines 1–3) determines all Placements that satisfy software, hardware, IoT and network QoS requirements of the application by means of `gFogBrain/4`, along with the associated hourly deployment Cost, energy consumption E and carbon emissions c (line 2). The obtained placements are sorted by increasing estimated carbon footprint, cost, and energy consumption, considered in this order of priority³ (line 3).

² Due to space limitations, we only show the main predicates of **GFogBrain**. Full code is open-sourced at <https://github.com/di-unipi-socc/fogbrainx/tree/main/green>.

³ By suitably rearranging output tuples, it is possible to prioritise differently among the estimated metrics. For instance, the order `(Cost,E,C,P)` at line 2 would give priority to cost over energy consumption and carbon emissions.

Predicate `gFogBrain/4` (lines 4–6) exploits the `placement/2` (line 5) and the `allocatedResources/2` (line 6) predicates of [10] (see Appendix A) to determine a placement that satisfies software, hardware, IoT and network QoS requirements of a given application, and the associated hardware and bandwidth in use, respectively. Then, Energy consumption and Carbon emissions associated to the placement are computed via the `footprint/4` predicate (lines 6, 7–11).

Based on the deployment nodes used by `P` (line 8), the predicate `footprint/4` computes hardware- (line 9) and network-related (line 10) energy consumption and carbon emissions and sums them, respectively (line 11). `GFogBrain` employs an extended version of the model from Kelly et al. [16] to associate an estimate of energy consumption to a piece of computation running on a given node. The energy consumption E_s of a service s running on a node n is computed as

$$E_s = E_n \cdot \text{PUE}_n \quad [\text{kWh}] \quad (1)$$

where E_n is the energy consumption caused by s at n (in kWh) and PUE_n is the PUE of n . As aforementioned, E_n is a (non-linear) function of the current node load. For each node `N` involved in placement `P`, `hardwareFootprint/4` (line 9, 12–17) exploits `hardwareEnergy/4` (line 14, 18–23) to first retrieve the node load `OldL` before placing the services in placement `P`, and the associated energy consumption `OldE` (line 20). Then, it retrieves the node load `NewL` after placing the services as per `P` (line 21), and computes the associated energy consumption `NewE` (line 22). The difference between `NewE` and `OldE`, multiplied by the PUE of `N`, estimates the Energy consumption of `P` on node `N` as per Eq. (1) (line 23).

Based on this, `GFogBrain` also estimates the associated carbon emissions. To this end, extending [16], we consider the case in which multiple energy sources are combined at node n – each with an associated mix percentage p_1, \dots, p_k such that $\sum_i p_i = 1$ – producing μ_1, \dots, μ_k emissions, respectively. Predicate `hardwareEmissions/3` (line 15, 24–27) recursively scans the energy mix declared for node `N` and computes carbon emissions as

$$I_s = E_s \cdot \sum_i p_i \mu_i \quad [\text{kgCO}_2] \quad (2)$$

Finally, following the approach of [19], `networkFootprint/3` (lines 10, 28–31) estimates the carbon emissions to transmit traffic flows allocated by `P`. Transmitting 1MB of data over the Internet requires around 0.00008 kWh (`kWhPerMB/1`) [14] and the average *global carbon intensity* (`averageGCI/1`) of electricity is of 475 gCO₂/kWh [15]. Then, the network energy consumption E_N and carbon emissions I_N for transmitting M MB for one hour can be estimated as

$$E_N = 450 \cdot 0.00008 \cdot M \quad [\text{kWh}] \quad \text{and} \quad I_N = 0.475 \cdot E_N \quad [\text{kgCO}_2] \quad (3)$$

also considering that 1 Mbit/s = 450 MB/h. Eq.s (3) are computed at lines 30 and 31 of the code of Fig. 6, respectively.

Example 3. By querying `placements(lightsApp,Placements)` over the inputs of Examples 1 and 2, we obtain the two eligible placements for application `LightsApp` listed in Table 2, along with their estimated hourly carbon emissions, energy

```

1 placements(A,Placements) :-
2     findall((C,Cost,E,P), (gFogBrain(A,P,E,C), hourlyCost(P,Cost)), Ps),
3     sort(Ps,Placements).

4 gFogBrain(A,P,Energy,Carbon) :-
5     application(A,Services), placement(Services,P),
6     allocatedResources(P,Alloc), footprint(P,Alloc,Energy,Carbon).

7 footprint(P,(AllocHW,AllocBW),Energy,Carbon) :-
8     deploymentNodes(P,Nodes),
9     hardwareFootprint(Nodes,AllocHW,HWEnergy,HWCarbon),
10    networkFootprint(AllocBW,BWEnergy,BWCarbon),
11    Energy is HWEnergy + BWEnergy, Carbon is HWCarbon + BWCarbon.

12 hardwareFootprint([(N,HW)|Ns],AllocHW,Energy,Carbon) :-
13    hardwareFootprint(Ns,AllocHW,EnergyNs,CarbonNs),
14    hardwareEnergy(N,HW,AllocHW,EnergyN),
15    energySourceMix(N,Sources), hardwareEmissions(Sources,EnergyN,CarbonN),
16    Energy is EnergyN + EnergyNs, Carbon is CarbonN + CarbonNs.
17 hardwareFootprint([],_,0,0).

18 hardwareEnergy(N,HW,AllocHW,Energy):-
19    totHW(N,TotHW), pue(N,PUE),
20    OldL is 100 * (TotHW - HW) / TotHW, energyProfile(N,OldL,OldE),
21    findall(H,member((N,H),AllocHW),HWs), sum_list(HWs,PHW),
22    NewL is 100 * (TotHW - HW + PHW) / TotHW, energyProfile(N,NewL,NewE),
23    Energy is (NewE - OldE) * PUE.

24 hardwareEmissions([(P,S)|Srcs],Energy,Carbon) :-
25    hardwareEmissions(Srcs,Energy,CarbSrcs),
26    emissions(S,MU), CarbS is P * MU * Energy, Carbon is CarbS + CarbSrcs.
27 hardwareEmissions([],_,0).

28 networkFootprint(AllocBW,BWEnergy,BWCarbon) :-
29    findall(BW, member((_,_,BW),AllocBW), Flows), sum_list(Flows,TotBW),
30    kWhPerMB(K), BWEnergy is 450 * K * TotBW,
31    averageGCI(A), BWCarbon is A * BWEnergy.

```

Fig. 5. Main predicates of **GFogBrain**.

consumption, and cost. Based on those and on business considerations, application operators can then informedly decide whether to enact P_1 or P_2 . While P_1 saves more than 9% CO₂ emissions compared to P_2 , and consumes 5% less energy, it incurs in an 11% cost increase (i.e. +0.004 €/h \simeq +3 €/month). It is also possible to exploit **GFogBrain** to perform *what-if* analyses and to possibly evaluate *greener* infrastructure operators, thus improving on target metrics. \square

Table 2. Example placement results.

Id	Placement	Emissions	Cost	Energy Cons.
P_1	on(lightsDriver, edgenode), on(mlOptimiser, privateCloud)	0.29 kgCO ₂	0.0356 €/h	0.60 kWh
P_2	on(lightsDriver, accesspoint), on(mlOptimiser, privateCloud)	0.32 kgCO ₂	0.0316 €/h	0.63 kWh

3 Related Work

Much work targeted the problem of placing multi-service applications in Cloud-IoT computing scenarios, e.g. as surveyed in [5,18]. Only some works featured some aspects of energy-awareness but did not consider carbon footprint or relied on simple linear models for energy consumption (e.g. [2,17,23,25]). To the best of our knowledge, [1] is the first work including carbon emissions in the trade-off analyses to determine optimal Cloud-IoT application placements, via mixed integer linear programming. A limitation of [1] resides in the fact that it only considers linear energy consumption for infrastructure nodes. On the contrary, energy consumption is usually a non-linear function of a computational node load [19,24]. Last, [1] does not consider the possibility to estimate energy consumption based on combined sources, do not account for operational costs estimates, and require full knowledge of the physical network topology and employed routing algorithms, which is not always available in real scenarios.

Focussing on declarative approaches, Casadei et al. [9,21] devised a declarative approach to service coordination based on *aggregate computing*, managing opportunistic resources via a hybrid centralised/decentralised solution by relying on a self-organising peer-to-peer architecture to handle churn and mobility. We have exploited logic programming to assess the security and trust levels of application placements [12], and to determine them [10] also in Osmotic computing settings [11]. Finally, we very recently proposed a fully decentralised solution to write and enforce QoS-aware application management policies written in Prolog [6]. None of those declarative solutions, however, considers energy consumption nor carbon emissions, as **GFogBrain** does.

4 Concluding Remarks

In this article, we have presented a declarative methodology and its prototype, **GFogBrain**, to determine eligible multiservice application placements and to estimate their carbon emissions, energy consumption and operational costs. The prototype allows application operators to determine application placements that satisfy all software, hardware, IoT, and network QoS constraints, and to informedly decide on the best trade-off placement considering its estimated impact on the environment and its deployment operational costs, which oftentimes represent contrasting objectives to optimise at once.

Future work includes improving **GFogBrain** with *continuous reasoning* (as in [10]) and assessing it via simulation or over Cloud-IoT testbeds, also employing different formulas to estimate energy consumption and carbon emissions (including an amortised estimate of the deployment hardware *embedded carbon*).

References

1. Aldossary, M., Alharbi, H.A.: Towards a green approach for minimizing carbon emissions in fog-cloud architecture. *IEEE Access* (2021)
2. Barcelo, M., Correa, A., Llorca, J., Tulino, A.M., Vicario, J.L., Morell, A.: Iot-cloud service optimization in next generation smart environments. *IEEE Journal on Selected Areas in Communications* **34**(12), 4077–4090 (2016)
3. Belkhir, L., Elmeligi, A.: Assessing ict global emissions footprint: Trends to 2040 & recommendations. *Journal of cleaner production* **177**, 448–463 (2018)
4. Bisicchia, G., Forti, S., Brogi, A.: Declarative goal mediation in smart environments. In: 2021 IEEE International Conference on Smart Computing (SMARTCOMP). pp. 389–391 (2021). <https://doi.org/10.1109/SMARTCOMP52413.2021.00079>
5. Brogi, A., Forti, S., Guerrero, C., Lera, I.: How to Place Your Apps in the Fog - State of the Art and Open Challenges. *Softw. Pract. Exp.* **50**(5), 719–740 (2020). <https://doi.org/10.1002/spe.2766>
6. Brogi, A., Forti, S., Guerrero, C., Lera, I.: Towards declarative decentralised application management in the fog. In: ISSRE Workshops. pp. 223–230 (2020). <https://doi.org/10.1109/ISSREW51248.2020.00077>
7. Brogi, A., Forti, S., Ibrahim, A., Rinaldi, L.: Bonsai in the fog: An active learning lab with fog computing. In: 2018 Third International Conference on Fog and Mobile Edge Computing (FMEC). pp. 79–86. IEEE (2018)
8. Calero, C., Piattini, M.: *Green in software engineering*, vol. 3. Springer (2015)
9. Casadei, R., Viroli, M.: Coordinating computation at the edge: a decentralized, self-organizing, spatial approach. In: FMEC 2019. pp. 60–67 (2019). <https://doi.org/10.1109/FMEC.2019.8795355>, <https://doi.org/10.1109/FMEC.2019.8795355>
10. Forti, S., Brogi, A.: Continuous reasoning for managing next-gen distributed applications. In: Ricca, F., Russo, A., Greco, S., Leone, N., Artikis, A., Friedrich, G., Fodor, P., Kimmig, A., Lisi, F.A., Maratea, M., Mileo, A., Riguzzi, F. (eds.) *Proceedings 36th International Conference on Logic Programming (Technical Communications), ICLP Technical Communications 2020, (Technical Communications) UNICAL, Rende (CS), Italy, 18-24th September 2020. EPTCS, vol. 325, pp. 164–177 (2020)*. <https://doi.org/10.4204/EPTCS.325.22>, <https://doi.org/10.4204/EPTCS.325.22>
11. Forti, S., Brogi, A.: Declarative osmotic application placement. In: Polyvyanyy, A., Rinderle-Ma, S. (eds.) *Advanced Information Systems Engineering Workshops - CAiSE 2021 International Workshops, Melbourne, VIC, Australia, June 28 - July 2, 2021, Proceedings. Lecture Notes in Business Information Processing, vol. 423, pp. 177–190. Springer (2021)*. https://doi.org/10.1007/978-3-030-79022-6_15, https://doi.org/10.1007/978-3-030-79022-6_15
12. Forti, S., Ferrari, G.L., Brogi, A.: Secure Cloud-Edge Deployments, with Trust. *Future Gener. Comput. Syst.* **102**, 775–788 (2020). <https://doi.org/10.1016/j.future.2019.08.020>
13. Greenpeace: Clicking green. who is winning the race to build a green internet? (2017)
14. IEA: The carbon footprint of streaming video: fact-checking the headlines. <https://www.iea.org/commentaries/the-carbon-footprint-of-streaming-video-fact-checking-the-headlines>

15. IEA: Global energy & co2 status report 2019. <https://www.iea.org/reports/global-energy-co2-status-report-2019/>
16. Kelly, C., Mangina, E., Ruzelli, A.: Putting a co2 figure on a piece of computation. In: 11th International Conference on Electrical Power Quality and Utilisation. pp. 1–7 (2011). <https://doi.org/10.1109/EPQU.2011.6128960>
17. Koprass, B., Idzikowski, F., Chen, W.C., Wang, T.J., Chou, C.T., Bogucka, H.: Latency-aware virtual network embedding using clusters for green fog computing. In: 2020 IEEE Globecom Workshops (GC Wkshps. pp. 1–7. IEEE (2020)
18. Mahmud, R., Ramamohanarao, K., Buyya, R.: Application Management in Fog Computing Environments: A Taxonomy, Review and Future Directions. *ACM Comput. Surv.* **53**(4) (2020)
19. Microsoft: The principles of sustainable software engineering. <https://docs.microsoft.com/en-us/learn/modules/sustainable-software-engineering-overview/>
20. Mourão, B.C., Karita, L., do Carmo Machado, I.: Green and sustainable software engineering - a systematic mapping study. *ACM* (2018)
21. Pianini, D., Casadei, R., Viroli, M., Natali, A.: Partitioned integration and coordination via the self-organising coordination regions pattern. *Future Gener. Comput. Syst.* **114**, 44–68 (2021). <https://doi.org/10.1016/j.future.2020.07.032>
22. Sarkar, S., Misra, S.: Theoretical modelling of fog computing: a green computing paradigm to support iot applications. *Iet Networks* **5**(2), 23–29 (2016)
23. Souza, V.B., Masip-Bruin, X., Marín-Tordera, E., Ramírez, W., Sánchez, S.: Towards distributed service allocation in fog-to-cloud (f2c) scenarios. In: 2016 IEEE global communications conference (GLOBECOM). pp. 1–6. IEEE (2016)
24. Xiao, Y., Zhang, Y., Kaku, I., Kang, R., Pan, X.: Electric vehicle routing problem: A systematic review and a new comprehensive model with nonlinear energy recharging and consumption. *Renewable and Sustainable Energy Reviews* **151**, 111567 (2021)
25. Yu, Y., Bu, X., Yang, K., Wu, Z., Han, Z.: Green large-scale fog computing resource allocation using joint benders decomposition, dinkelbach algorithm, admm, and branch-and-bound. *IEEE Internet of Things Journal* **6**(3), 4106–4117 (2018)

Appendix A – Code of `placement/2`

For the sake of the Reviewers only, we report hereinafter the code of `placement/4` from [10], exploited by predicate `gFogBrain/4` (Fig. 6).

```

1 placement(Services,P) :- placement(Services, [], ([],[]), P).
2 placement([S|Ss],P,(AllocHW,AllocBW),Placement) :-
3   nodeOk(S,N,P,AllocHW), linksOk(S,N,P,AllocBW),
4   placement(Ss,[on(S,N)|P],(AllocHW,AllocBW),Placement).
5 placement([],P,-,P).
6
6 nodeOk(S,N,P,AllocHW) :-
7   service(S,SWReqs,HWRqs,IoTReqs),
8   node(N,SWCaps,HWCaps,IoTCaps),
9   swReqsOk(SWReqs,SWCaps),
10  thingReqsOk(IoTReqs,IoTCaps),
11  hwOk(N,HWCaps,HWRqs,P,AllocHW).
12
12 swReqsOk(SWReqs, SWCaps) :- subset(SWReqs, SWCaps).
13
13 thingReqsOk(TReqs, TCaps) :- subset(TReqs, TCaps).
14
14 hwOk(N,HWCaps,HWRqs,P,AllocHW) :-
15   findall(HW,member((N,HW),AllocHW),HWS),
16   sum_list(HWS,CurrAllocHW),
17   findall(HW,(member(on(S1,N),P),service(S1,-,HW,-)),OkHWS),
18   sum_list(OkHWS,NewAllocHW),
19   hwTh(T),HWCaps >= HWRqs + T - CurrAllocHW + NewAllocHW.
20
20 linksOk(S,N,P,AllocBW) :-
21   findall((N1N2,ReqLat),distinct(relevant(S,N,P,N1N2,ReqLat)),N2Ns),
22   latencyOk(N2Ns),
23   findall(N1N2,distinct(member((N1N2,ReqLat),N2Ns)),N1N2s),
24   bwOk(N1N2s,AllocBW,[on(S,N)|P]).
25
25 latencyOk([(N1,N2),ReqLat]|N2Ns]):-
26   link(N1,N2,FeatLat,-),FeatLat <= ReqLat,latencyOk(N2Ns).
27 latencyOk([]).
28
28 bwOk([(N1,N2)|N2Ns],AllocBW,P) :-
29   link(N1,N2,-,FeatBW),
30   findall(BW,member((N1,N2,BW),AllocBW),BWs),
31   sum_list(BWs,CurrAllocBW),
32   findall(BW,s2sOnN1N2((N1,N2),P,BW),OkBWs),
33   sum_list(OkBWs,OkAllocBw),
34   bwTh(T),FeatBW >= OkAllocBw - CurrAllocBW + T,
35   bwOk(N2Ns,AllocBW,P).
36 bwOk([],-,_-).
37
37 relevant(S,N,P,(N,N2),L) :- s2s(S,S2,L,-),member(on(S2,N2),P),dif(N,N2).
38 relevant(S,N,P,(N1,N),L) :- s2s(S1,S,L,-),member(on(S1,N1),P),dif(N1,N).
39
39 s2sOnN1N2((N1,N2),P,B) :-
40   s2s(S3,S4,-,B),member(on(S3,N1),P),member(on(S4,N2),P).
41
41 allocatedResources(P,(AllocHW,AllocBW)) :-
42   findall((N,HW),(member(on(S,N),P),service(S,-,HW,-)),AllocHW),
43   findall((N1,N2,BW),n2n(P,N1,N2,BW),AllocBW).
44 n2n(P,N1,N2,ReqBW) :-
45   s2s(S1,S2,-,ReqBW),member(on(S1,N1),P),member(on(S2,N2),P),dif(N1,N2).

```

Fig. 6. Declarative placement strategy of [10].